```
(*
```

Programma di esempio di calcolo di mode–matching usando il metodo delle matrici di propagazione (ABCD)

```
*)

(* si lavora in mm  *)

(* Matrici di: lente sottile di focale f (Mf), propagazione lunga d (Md),
passaggio attraverso una superficie sferica di raggio R,
che delimita due mezzi con indici di rifrazione n1 e n2 (MnR)  *)
```

$$\text{Mf[f\_]} := \begin{pmatrix} 1 & 0 \\ -1/f & 1 \end{pmatrix};\ \text{Md[d\_]} := \begin{pmatrix} 1 & d \\ 0 & 1 \end{pmatrix};\ \text{MnR[n1\_, n2\_, r\_]} := \begin{pmatrix} 1 & 0 \\ (n2-n1)/r & 1 \end{pmatrix}$$

```
(* matrici esatte per lenti piano-convessa e convesso-piana,
di raggio r, vetro con indice di rifrazione nrf, spessore t   *)

Mpc[nrf_, t_, r_] := Dot[MnR[1, nrf, -r], Md[t / nrf]]
Mcp[nrf_, t_, r_] := Dot[Md[t / nrf], MnR[1, nrf, -r]]

(* Si definiscono gli alementi ABCD della matrice totale del sistema 'Mtot' (che costruiremo dopo) *)

mA := Mtot[[1, 1]];
mB := Mtot[[1, 2]];
mC := Mtot[[2, 1]];
mD := Mtot[[2, 2]];
```

(* Si suppone di partire da un fuoco del fascio, dove il waist è 'win',
e di arrivare sul fuoco, con waist finale 'wout'. Il parametro di fascio q è quindi
immaginario puro sia in ingresso che in uscita. Valgono quindi le relazione $BD+\frac{AC\ \pi^2\ win^4}{\lambda^2}=0$,
dove λ è la lunghezza d'onda, e $wout=win*\sqrt{\frac{A}{D}}$ (dimostrate in coda a questo foglio). Queste due
equazioni definiscono completamente il problema, che avrà quindi due incognite libere. *)

$$\text{wout} := \text{win} * \sqrt{\frac{\text{mA}}{\text{mD}}}$$

$$\text{equazione} := \left(\text{mB} * \text{mD} + \frac{\pi\text{^}2 * \text{win}\text{^}4 * \text{mA} * \text{mC}}{\lambda\text{^}2} == 0\right)$$

```
(*      I  ESEMPIO: Adattamento del modo con due lenti.             *)
(*  Il sistema ottico complessivo è formato da una propagazione lunga 'd0',
una lente di focale 'f1', una distanza 'd1',
una lente di focale 'f2', un tratto finale lungo 'd2'.   *)

 Mtot := Dot[Md[d2], Mf[f2], Md[d1], Mf[f1], Md[d0]]

( * Esempio 1 a : conosciamo il waist in ingresso e quello che vogliamo in uscita ( ' wf ' ),  le focali delle due lenti
   e la lunghezza totale del sistema ottico ( ' dtot ' ) . Le incognite sono le posizioni delle lenti.  *)

(*  Definizione dei parametri noti  *)

λ = 0.78 × 10^-3;
win = 0.8;
wf = 0.2;
f1 = 100;
f2 = 100;
dtot = 1500;
(* Soluzione delle equazioni del sistema  *)

Solve[{equazione, wout == wf, d0 + d1 + d2 == dtot}, {d0, d1, d2}]

Clear[λ, win, wf, f1, f2, dtot]
{{d0 → 1993.63, d1 → 189.602, d2 → -683.229},
 {d0 → 101.28 - 2580.88 i, d1 → 1297.46 + 2745.13 i, d2 → 101.257 - 164.25 i},
 {d0 → 101.28 + 2580.88 i, d1 → 1297.46 - 2745.13 i, d2 → 101.257 + 164.25 i},
 {d0 → 109.246, d1 → 1281.51, d2 → 109.246},
 {d0 → 640.754, d1 → 218.493, d2 → 640.754}, {d0 → 550.479, d1 → 215.473, d2 → 734.048}}
(* Vengono trovate 6 soluzioni. Di queste 2 sono non fisiche (compaiono parti immaginarie) ed una
 propone una distanza negativa. Rimangono 3 soluzioni valide, tra le quali si può scegliere  *)

(*      II  ESEMPIO: Adattamento del modo a quello di una cavità, usando tre lenti.            *)
(*  Uso di tre lenti sottili per adattare il modo a quello di una cavità ottica lunga 'lung',
composta da uno specchio di ingresso piano-concavo di fused silica (indice di rifrazione 1.445),
raggio di curvatura 'raggio', spessore 6.35mm  *)

Mtot := Dot[Md[lung], Mpc[1.445, 6.35, -raggio], Md[d3], Mf[f3], Md[d2], Mf[f2], Md[d1], Mf[f1], Md[d0]]
```

```
(*  Esempio 2a: conosciamo il waist in ingresso e calcoliamo quello della cavità ('wcav');
fissiamo le focali delle tre lenti e le prime due distanze. Le
 incognite sono la posizione dell aterza lente e quell adella cavità.  *)

f1 = 300;
f2 = 50;
f3 = 125;
d0 = 60;
d1 = 300;
win = 0.4;
λ = 0.78 × 10^-3;
lung = 10;
raggio = 200;

wcav := Sqrt[λ Sqrt[lung * (raggio - lung)] / π];
N[wcav]

0.104031

Solve[{equazione, wout == wcav}, {d2, d3}]

{{d2 → 100.158, d3 → -518.147}, {d2 → 88.9581, d3 → -66.3601},
 {d2 → 259.607, d3 → 280.162}, {d2 → 207.534, d3 → 300.138}}
```

```
(* Esempio 2b: si trova il fuoco finale (posizione e waist) al variare della posizione della terza lente,
con d2 variabile tra 200mm e 300mm
 (ovvero, intorno alla posizione suggerita precedentemente dalla terza soluzione)  *)

xout := d3 /. Solve[equazione, d3]
wf := wout /. {d3 → xout}

Plot[(d0 + d1 + d2 + xout), {d2, 200, 300}, PlotRange → All]
Plot[(wf), {d2, 200, 300}, PlotRange → All]
```






```
(* Esempio 2c: si trova l'ultima lente fissando lo spazio disponibile.
    Puo' servire per trovare valori ragionevoli di focale. *)

Clear[f3]
dtot = 600;
Solve[{equazione, wout == wcav, d2 + d3 == dtot}, {f3, d2, d3}]

{{f3 → 139.409, d2 → 284.381, d3 → 315.619}, {f3 → 144.696, d2 → 233.371, d3 → 366.629},
 {f3 → -704.619, d2 → -418.903, d3 → 1018.9}, {f3 → -4489.15, d2 → -1289.65, d3 → 1889.65}}
```

```
(* NSolve è più veloce di Solve, e sufficientemente accurato,
ma può essere che non trovi tutte le soluzioni. *)

Clear[f3]
dtot = 600;
NSolve[{equazione, wout == wcav, d2 + d3 == dtot}, {f3, d2, d3}]
```

```
{{f3 → -4489.15, d2 → -1289.65, d3 → 1889.65}, {f3 → 139.409, d2 → 284.381, d3 → 315.619}}
```

```
(* Esempio 2d: Si è visto che f3=
 139.4 e f3=144.7 darebbero buoni risultati. Si ottimizza la posizione delle due ultime lenti,
sempre fissando lo spazio disponibile,
con una lente disponibile (f3=125) vicina ai risultati precedenti.   *)

Clear[d1]
f3 = 125;
dtot := 900;
NSolve[{equazione, wout == wcav, d1 + d2 + d3 == dtot}, {d1, d2, d3}]
```

```
{{d1 → -326.151, d2 → 187.923, d3 → 1038.23},
 {d1 → 314.686 + 125.967 i, d2 → 438.71 - 195.233 i, d3 → 146.603 + 69.2659 i},
 {d1 → 314.686 - 125.967 i, d2 → 438.71 + 195.233 i, d3 → 146.603 - 69.2659 i},
 {d1 → 346.999, d2 → 262.207, d3 → 290.794}}
```

(* **DIMOSTRAZIONE DELLE RELAZIONI** $BD+\frac{AC\ \pi^2\ win^4}{\lambda^2}=0$ **e** $wout=win*\sqrt{\frac{A}{D}}$ *)

(*
La relazione che descrive la trasformazione
del parametro complesso q del fascio operata dal sistema ottico è
(1) $qout = \frac{A\ qin + B}{C\ qin + D}$
Per ipotesi,
in ingresso siamo nel fuoco del fascio. qin è quindi immaginario puro e si può scrivere come
$qin = -iz_{in}$ dove $z_{in}=$
$\frac{\pi\ win^2}{\lambda}$ è il parametro confocale (reale). Lo stesso vale in uscita. La (1) si può quindi scrivere come
(2) $-iz_{out} = \frac{-iA\ z_{in} + B}{-iC\ z_{in} + D}$
che, razionalizzata (tenendo conto che gli elementi di matrice sono reali), diventa
(3) $-iz_{out} = \frac{BD + (z_{in})^2 AC - i\ z_{in}\ (AD - BC)}{(C\ z_{in})^2 + D^2}$
La richiesta Re[qout]=0 comporta quindi
(4) $BD + (z_{in})^2 AC = 0$
che, sostituendo l'espressione di z_{in}, si scrive
(5) $BD+\frac{AC\ \pi^2\ win^4}{\lambda^2}=0$
Riprendendo la (3) ed utilizzando la (4), abbiamo
(6) $z_{out} = \frac{z_{in}\ (AD - BC)}{(C\ z_{in})^2 + D^2}$
Utilizzando la (4), il denominatore si può scrivere come $\frac{D}{A}(AD-BC)$ e,
semplificando col numeratore ed utilizzando le espressioni di z_{in} e z_{out}, si ottiene
(7) $wout=win*\sqrt{\frac{A}{D}}$
Notiamo che (AD-BC) è il determinante della matrice del sistema ottico, e vale sempre 1.
*)